Q3 TUTORIAL

Related Tech Notes · Related Guides · Functions · URL

Quantum Teleportation

'PDATED IN 13

See also Section 4.1 of the Quantum Workbook (Springer, 2022) .

Quantum teleportation is a communication protocol that sends quantum information to a distant party by utilizing a pre-shared pair of entangled particles (Bennett *et al*., 1993). The protocol has attracted public interest since it resembles (fictional) teleportation. In science fiction, (hypothetical) teleportation instantly transports matter or energy across space. Quantum teleportation does not transfer physical objects but only quantum information. Quantum teleportation is also conceptually distinguished from telefax. Telefax makes a copy of the document and sends the duplicated copy while the original document remains at its origin. In quantum teleportation, a quantum state is transferred and disappears from the original system.

Quantum teleportation is one of the first quantum information protocols that vividly illustrates the significance of quantum entanglement as a valuable resource. In this section, we discuss the physical principle behind quantum teleportation and demonstrate the protocol. Another closely related quantum communication protocol is the so-called superdense coding (Bennett and Wiesenr, 1992).

Quantum teleportation is not only interesting in its own right but also used as part of various quantum algorithms. Interestingly, even a computational model based solely on quantum teleportation has been proposed as well (Gottesman and Chuang, 1999; Nielsen, 2003; Leung, 2004).

QuantumCircuit	Circuit representation of elementary gates.
CNOT	Two–qubit controlled–NOT gate.
Hadamard	The Hadamard gate.
BellState	Returns a list of the four Bell states.
Matrix	Converts QuantumCircuit and operator (or Ket) expressions to the matrix representation in the standard basis.
ExpressionFor	Converts the matrix representation to the corresponding operator or state–vector expression.

Functions related to the quantum teleportation protocol.

Make sure that the Q3 package is loaded to use the demonstrations in this document.

```
In[1]:= << Q3`
```

Throughout this document, symbol S will be used to denote qubits and Pauli operators on them. Similarly, symbol c will be used to denote complex-valued coefficients.

```
In[2]:= Let[Qubit, S]
        Let[Complex, c]
```

©1988–2022 Wolfram Research, Inc. All rights reserved. http://reference.wolfram.com/language

Synopsis

Suppose that Bob wants to send one bit of quantum information, say, $\left|\psi\right\rangle_C = \left|0\right\rangle_C c_0 + \left|1\right\rangle_C c_1$ stored in Charlie's qubit, to Alice residing in a place far away from Bob (and Charlie). There is no quantum communication channel operating between Alice and Bob to directly send the quantum state $\left|\psi\right\rangle$ through. Fortunately, however, there is a classical commutation channel.

It is important to note that the quantum state $\left|\psi\right\rangle$ is unknown. The no-cloning theorem prevents a naive approach such as copying and transmitting the quantum state.

Simple Theory

In[4]:=
```
<< Q3`
```

In[376]:=
```
Let[Qubit, S]
Let[Complex, c]
```

To teleport a quantum state, Alice and Bob need to share an entangled pair of qubits in advance. Suppose that the shared pair of qubits is in the state $\left|\beta_0\right\rangle_{AB}$, which is one of the four Bell states,

$$\left|\beta_0\right\rangle := \frac{|00\rangle+|11\rangle}{\sqrt{2}}, \quad \left|\beta_1\right\rangle := \frac{|01\rangle+|10\rangle}{\sqrt{2}}, \quad \left|\beta_3\right\rangle := \frac{|01\rangle-|10\rangle}{\sqrt{2}}, \quad \left|\beta_3\right\rangle := \frac{|00\rangle-|11\rangle}{\sqrt{2}}.$$

In[378]:=
```
bs = BellState[S@{1, 2}];
bs // LogicalForm
```

Out[379]=

$$\left\{\frac{\left|0_{S_1}0_{S_2}\right\rangle + \left|1_{S_1}1_{S_2}\right\rangle}{\sqrt{2}}, \frac{\left|0_{S_1}1_{S_2}\right\rangle + \left|1_{S_1}0_{S_2}\right\rangle}{\sqrt{2}}, \frac{\left|0_{S_1}1_{S_2}\right\rangle - \left|1_{S_1}0_{S_2}\right\rangle}{\sqrt{2}}, \frac{\left|0_{S_1}0_{S_2}\right\rangle - \left|1_{S_1}1_{S_2}\right\rangle}{\sqrt{2}}\right\}$$

Then, with the quantum state stored in qubit C, that is,

$$\left|\psi\right\rangle_C = \left|0\right\rangle_C c_0 + \left|1\right\rangle_C c_1 ,$$

In[380]:=
```
msg = Ket[] * c[0] + Ket[S[3] -> 1] * c[1];
LogicalForm[msg]
```

Out[381]=

$$c_0 \left|0_{S_3}\right\rangle + c_1 \left|1_{S_3}\right\rangle$$

the initial state $\left|\Psi\right\rangle$ of the overall three qubits at the beginning of the procedure is given by

$$\left|\Psi\right\rangle = \left|\beta_0\right\rangle_{AB} \otimes \left|\psi\right\rangle_C = \frac{|000\rangle c_0 + |110\rangle c_0 + |001\rangle c_1 + |111\rangle c_1}{\sqrt{2}}.$$

In[382]:=
```
in = bs[[1]] ** msg;
LogicalForm[in]
```

Out[383]=

$$\frac{c_0 \left|0_{S_1}0_{S_2}0_{S_3}\right\rangle}{\sqrt{2}} + \frac{c_1 \left|0_{S_1}0_{S_2}1_{S_3}\right\rangle}{\sqrt{2}} + \frac{c_0 \left|1_{S_1}1_{S_2}0_{S_3}\right\rangle}{\sqrt{2}} + \frac{c_1 \left|1_{S_1}1_{S_2}1_{S_3}\right\rangle}{\sqrt{2}}$$

©1988–2022 Wolfram Research, Inc. All rights reserved. http://reference.wolfram.com/language

Recall that Bob has qubits *B* and *C* at his hands.

In[26]:=

```
ProductForm[in, {S@1, S@{2, 3}}]
```

Out[26]=

$$\frac{c_0\,|0\rangle\,|00\rangle}{\sqrt{2}}+\frac{c_1\,|0\rangle\,|01\rangle}{\sqrt{2}}+\frac{c_0\,|1\rangle\,|10\rangle}{\sqrt{2}}+\frac{c_1\,|1\rangle\,|11\rangle}{\sqrt{2}}$$

Rewriting the parts consisting of qubits B and C in the Bell basis, the basis consisting of the four **Bell states**, leads to

$$|\Psi\rangle = \frac{1}{2}\left(|0\rangle\, c_0 + |1\rangle\, c_1\right)_A \otimes |\beta_0\rangle_{B\,C} + \frac{1}{2}\left(|0\rangle\, c_1 + |1\rangle\, c_0\right)_A \otimes |\beta_1\rangle_{B\,C} + \frac{1}{2}\left(|0\rangle\, c_1 - |1\rangle\, c_0\right)_A \otimes |\beta_2\rangle_{B\,C} + \frac{1}{2}\left(|0\rangle\, c_0 - |1\rangle\, c_1\right)_A \otimes |\beta_3\rangle_{B\,C}$$

In[390]:=

```
prj = Dyad[#, #, S@{2, 3}] & /@ BellState[S@{2, 3}];
cmp = prj ** in;
cmp // KetFactor // TableForm
```

Out[392]//TableForm=

$$\frac{\left(c_0\,|0_{S_1}\rangle + c_1\,|1_{S_1}\rangle\right)\otimes\left(|0_{S_2}0_{S_3}\rangle + |1_{S_2}1_{S_3}\rangle\right)}{2\,\sqrt{2}}$$
$$\frac{\left(c_1\,|0_{S_1}\rangle + c_0\,|1_{S_1}\rangle\right)\otimes\left(|0_{S_2}1_{S_3}\rangle + |1_{S_2}0_{S_3}\rangle\right)}{2\,\sqrt{2}}$$
$$-\frac{\left(c_1\,|0_{S_1}\rangle - c_0\,|1_{S_1}\rangle\right)\otimes\left(-|0_{S_2}1_{S_3}\rangle + |1_{S_2}0_{S_3}\rangle\right)}{2\,\sqrt{2}}$$
$$\frac{\left(c_0\,|0_{S_1}\rangle - c_1\,|1_{S_1}\rangle\right)\otimes\left(|0_{S_2}0_{S_3}\rangle - |1_{S_2}1_{S_3}\rangle\right)}{2\,\sqrt{2}}$$

The crucial point here is that the state of *A* in the first term is identical to the original quantum state $|\psi\rangle$, and those in the rest are also closely related to $|\psi\rangle$ by the Pauli operators. More explicitly, one can rewrite the total state vector as

$$|\Psi\rangle = |\psi\rangle_A \otimes |\beta_0\rangle_{B\,C} + \left(X\,|\psi\rangle\right)_A \otimes |\beta_1\rangle_{B\,C} + \left(i\,Y\,|\psi\rangle\right)_A \otimes |\beta_2\rangle_{B\,C} + \left(Z\,|\psi\rangle\right)_A \otimes |\beta_3\rangle_{B\,C}.$$

In[30]:=

```
out = {1, S[1, 1], I * S[1, 2], S[1, 3]} ** cmp;
KetFactor[out] // TableForm
```

Out[31]//TableForm=

$$\frac{\left(c_0\,|0_{S_1}\rangle + c_1\,|1_{S_1}\rangle\right)\otimes\left(|0_{S_2}0_{S_3}\rangle + |1_{S_2}1_{S_3}\rangle\right)}{2\,\sqrt{2}}$$
$$\frac{\left(c_0\,|0_{S_1}\rangle + c_1\,|1_{S_1}\rangle\right)\otimes\left(|0_{S_2}1_{S_3}\rangle + |1_{S_2}0_{S_3}\rangle\right)}{2\,\sqrt{2}}$$
$$\frac{\left(c_0\,|0_{S_1}\rangle + c_1\,|1_{S_1}\rangle\right)\otimes\left(-|0_{S_2}1_{S_3}\rangle + |1_{S_2}0_{S_3}\rangle\right)}{2\,\sqrt{2}}$$
$$\frac{\left(c_0\,|0_{S_1}\rangle + c_1\,|1_{S_1}\rangle\right)\otimes\left(|0_{S_2}0_{S_3}\rangle - |1_{S_2}1_{S_3}\rangle\right)}{2\,\sqrt{2}}$$

Now Bob performs a measurement on the two qubits *B* and *C*, owned by Bob, in the Bell basis. The measurement will yield outcomes $\mu = 0, 1, 2, 3$ and will collapse the total state vector into the corresponding term, $\left(S^\mu\,|\psi\rangle\right)_A \otimes |\beta_\mu\rangle_{B\,C}$, where $S^0 = I$, $S^1 = X$, $S^2 = i\,Y$, and $S^3 = Z$. The remaining task is for Bob to inform Alice of the outcome μ so that Alice recover the desired state $|\psi\rangle$ by operating the inverse operator S^μ on her qubit. The information regarding the measurement outcome amounts to two bits, and it requires only a classical channel for transmission.

In short, the quantum teleportation protocol consists of the following steps:

1. Alice and Bob generate an entangled pair of qubits, *A* and *B*, and share the pair between them. This can be done anytime before the procedure actually starts. Bob prepares a quantum state to send in a separate qubit *C*.
2. Bob makes a Bell measurement, i.e., the measurement in the Bell basis, on his two qubits *B* and *C*.
3. Bob sends the two-bit information regarding the measurement outcome to Alice through a classical communication channel.
4. Alice operates a proper inverse operator to recover the desired quantum state.

©1988–2022 Wolfram Research, Inc. All rights reserved. http://reference.wolfram.com/language

Preliminaries

In[32]:=

```
<< Q3`
```

In[33]:=

```
Let[Qubit, S]
Let[Complex, c]
```

In the quantum teleportation protocol, two essential parts are sharing entanglement and performing the Bell measurement. Before we demonstrate an implementation of the quantum teleportation protocol, here we summarize quantum circuit models of the two tasks.

Entangler Quantum Circuit

To establish an entanglement between two qubits initially in a product state, one needs to interact them. In the quantum circuit model, the interaction is achieve by the CNOT gate. Specifically, one can use the feature of the CNOT gate that copies the computational basis states of the control qubit to the target qubit,

$$(|0\rangle c_0 + |1\rangle c_1) \otimes |0\rangle = |00\rangle c_0 + |10\rangle c_1 \mapsto |00\rangle c_0 + |11\rangle c_1.$$

The linear superposition of the control qubit may be built by the Hadamard gate (a general form of linear superposition can be built with rotation gates).

In[156]:=

```
qc = QuantumCircuit[S[1, 6], CNOT[S[1], S[2]]]
```

Out[156]=



Get the four Bell states applying the above quantum circuit to the product basis.

In[160]:=

```
in = Basis[S@{1, 2}];
out = qc ** in;
Thread[in -> out] // LogicalForm // TableForm
```

Out[162]//TableForm=

$$|0_{S_1}0_{S_2}\rangle \to \frac{|0_{S_1}0_{S_2}\rangle}{\sqrt{2}} + \frac{|1_{S_1}1_{S_2}\rangle}{\sqrt{2}}$$

$$|0_{S_1}1_{S_2}\rangle \to \frac{|0_{S_1}1_{S_2}\rangle}{\sqrt{2}} + \frac{|1_{S_1}0_{S_2}\rangle}{\sqrt{2}}$$

$$|1_{S_1}0_{S_2}\rangle \to \frac{|0_{S_1}0_{S_2}\rangle}{\sqrt{2}} - \frac{|1_{S_1}1_{S_2}\rangle}{\sqrt{2}}$$

$$|1_{S_1}1_{S_2}\rangle \to \frac{|0_{S_1}1_{S_2}\rangle}{\sqrt{2}} - \frac{|1_{S_1}0_{S_2}\rangle}{\sqrt{2}}$$

Bell-Measurement Circuit

In quantum computers, a **measurement** usually means to measure the Pauli Z operator on different qubits. In other words, measurement is assumed to be performed independently on individual qubits in the computational basis, $\{|x\rangle : x = 0, 1, \ldots, 2n - 1\}$. (In Q3, measurement of arbitrary Pauli operators is supported; see the document of **Measurement**.)

To measure an arbitrary observable M, one needs to change the basis so that the eigenstates of M are mapped to the computational basis. Specifically, for the Bell measurement (a measurement in the basis of Bell states), the required basis change is equivalent to the inverse procedure of building an entanglement as discussed above. Therefore, the quantum circuit for the Bell measurement may be achieved simply by reversing the order of gates in the entangler circuit.

©1988–2022 Wolfram Research, Inc. All rights reserved. http://reference.wolfram.com/language

In[163]:=
```
qc = QuantumCircuit[CNOT[S[1], S[2]], S[1, 6]]
```
Out[163]=



In[165]:=
```
in = BellState[S@{1, 2}];
out = qc ** in;
Thread[in -> out] // LogicalForm // TableForm
```
Out[167]//TableForm=

$$\frac{|0_{S_1}0_{S_2}\rangle + |1_{S_1}1_{S_2}\rangle}{\sqrt{2}} \to |0_{S_1}0_{S_2}\rangle$$
$$\frac{|0_{S_1}1_{S_2}\rangle + |1_{S_1}0_{S_2}\rangle}{\sqrt{2}} \to |0_{S_1}1_{S_2}\rangle$$
$$\frac{|0_{S_1}1_{S_2}\rangle - |1_{S_1}0_{S_2}\rangle}{\sqrt{2}} \to |1_{S_1}1_{S_2}\rangle$$
$$\frac{|0_{S_1}0_{S_2}\rangle - |1_{S_1}1_{S_2}\rangle}{\sqrt{2}} \to |1_{S_1}0_{S_2}\rangle$$

We note that measurement outcomes (0, 0), (0, 1), (1, 1), and (1, 0) corresponds to the Bell states $|\beta_0\rangle$, $|\beta_1\rangle$, $|\beta_2\rangle$, and $|\beta_3\rangle$, respectively.

Implementation

In[38]:=
```
<< Q3`
```
In[39]:=
```
Let[Qubit, S]
Let[Complex, c]
```

Step 1. Alice and Bob generate and share an entangled pair using the quantum entangler circuit discussed above.

In[62]:=
```
qc1a = QuantumCircuit[LogicalForm[Ket[], S@{1, 2}],
   S[1, 6], CNOT[S[1], S[2]], "Invisible" → S[1.5], "Visible" -> S[3]];
Panel[qc1a]
```
Out[63]=



Bob also prepares the quantum state to send to Allice and stores it in Charlie's qubit *C*.

In[64]:=
```
msg = Ket[] × c[0] + Ket[S[3] → 1] × c[1];
msg // LogicalForm
```
Out[65]=

$c_0\ |0_{S_3}\rangle + c_1\ |1_{S_3}\rangle$

This is the quantum circuit that fulfills the tasks so far.

©1988–2022 Wolfram Research, Inc. All rights reserved. http://reference.wolfram.com/language

In[66]:=

```
qc1 = QuantumCircuit[{msg, "Label" -> Ket[ψ]}, qc1a, "Separator"]
```

Out[66]=



The overall state at this stage is as follows.

In[50]:=

```
out1 = Elaborate[qc1];
out1 // LogicalForm
out1 // KetFactor
```

Out[51]=

$$\frac{c_0 \left|0_{S_1}0_{S_2}0_{S_3}\right\rangle}{\sqrt{2}} + \frac{c_1 \left|0_{S_1}0_{S_2}1_{S_3}\right\rangle}{\sqrt{2}} + \frac{c_0 \left|1_{S_1}1_{S_2}0_{S_3}\right\rangle}{\sqrt{2}} + \frac{c_1 \left|1_{S_1}1_{S_2}1_{S_3}\right\rangle}{\sqrt{2}}$$

Out[52]=

$$\frac{\left(\left|0_{S_1}0_{S_2}\right\rangle + \left|1_{S_1}1_{S_2}\right\rangle\right) \otimes \left(c_0 \left|0_{S_3}\right\rangle + c_1 \left|1_{S_3}\right\rangle\right)}{\sqrt{2}}$$

Step 2. Bob performs a Bell measurement on his qubits. This can be done by reversing the entangler circuit as explained in the above.

In[53]:=

```
qc2a = QuantumCircuit[
   CNOT[S[2], S[3]], S[2, 6],
   Measurement[S[{2, 3}, 3]],
   "Visible" -> S[1], "Invisible" -> S[1.5]];
Panel[qc2a]
bs = BellState[S@{2, 3}];
ls = qc2a ** bs;
LogicalForm@Transpose@Thread[bs → ls] // TableForm
```

Out[54]=



Out[57]//TableForm=

$$\frac{\left|0_{S_2}0_{S_3}\right\rangle + \left|1_{S_2}1_{S_3}\right\rangle}{\sqrt{2}} \to \sqrt{2}\left|0_{S_2}0_{S_3}\right\rangle$$

$$\frac{\left|0_{S_2}1_{S_3}\right\rangle + \left|1_{S_2}0_{S_3}\right\rangle}{\sqrt{2}} \to \sqrt{2}\left|0_{S_2}1_{S_3}\right\rangle$$

$$\frac{\left|0_{S_2}1_{S_3}\right\rangle - \left|1_{S_2}0_{S_3}\right\rangle}{\sqrt{2}} \to \frac{\left|0_{S_2}1_{S_3}\right\rangle}{\sqrt{2}} + \frac{\left|1_{S_2}1_{S_3}\right\rangle}{\sqrt{2}}$$

$$\frac{\left|0_{S_2}0_{S_3}\right\rangle - \left|1_{S_2}1_{S_3}\right\rangle}{\sqrt{2}} \to \sqrt{2}\left|1_{S_2}0_{S_3}\right\rangle$$

The overall quantum circuit so far is as follows.

©1988–2022 Wolfram Research, Inc. All rights reserved. http://reference.wolfram.com/language

In[277]:=

```
qc2 = QuantumCircuit[qc1, qc2a, "Separator"]
```

Out[277]=



This is the state of the total system expected at this stage.

In[329]:=

```
normalizationRules = {c[0] * Conjugate[c[0]] + c[1] * Conjugate[c[1]] -> 1};
out2 = ExpressionFor[qc2] /. normalizationRules;
LogicalForm[out2, S@{1, 2, 3}] // KetFactor
```

Out[331]=

$\left(c_0 \left|0_{S_1}\right\rangle + c_1 \left|1_{S_1}\right\rangle\right) \otimes \left|0_{S_2} 0_{S_3}\right\rangle$

Step 3. Bob sends the measurement outcome to Alice through a classical channel.

Step 4. Alice applies a proper unitary gate on her qubit in accordance with Bob’s message. These two steps are simulated by a feedback control.

Combining all the steps, one can transmit a quantum state to a remote party without a quantum channel.

In[319]:=

```
qc = QuantumCircuit[qc2, ControlledU[S[2], S[1, 3]], ControlledU[S[3], S[1, 1]], "Invisible" → S[1.5]]
```

Out[319]=



This is the final state of the total system.

In[366]:=

```
out = ExpressionFor[qc] /. normalizationRules;
LogicalForm[out, S@{1, 2, 3}] // KetFactor
```

Out[367]=

$\left(c_0 \left|0_{S_1}\right\rangle + c_1 \left|1_{S_1}\right\rangle\right) \otimes \left|0_{S_2} 1_{S_3}\right\rangle$

Note that the Bell measurement outcomes are uniformed distributed, and the state resulting in Alice’s qubit is always the same as the original quantum state (`msg`) regardless of the measurement outcome.

©1988–2022 Wolfram Research, Inc. All rights reserved. http://reference.wolfram.com/language

In[384]:=

```
Timing[data = Table[ExpressionFor[qc] /. normalizationRules;
    Readout[S[{2, 3}, 3]], 300];]
Histogram3D[data, Ticks -> {{0, 1}, {0, 1}, Automatic}, AxesLabel → {"Z2", "Z3", "Counts"}]
```

Out[384]=

```
{9.54821, Null}
```

Out[385]=



Applications

In[4]:=

```
<< Q3`
```

In[5]:=

```
Let[Qubit, S]
Let[Complex, c]
```

As mentioned at the beginning, quantum teleportation is not only interesting in its own right but also used as part of various quantum algorithms. In fact, universal quantum computation based solely on quantum teleportation is possible (Gottesman and Chuang, 1999; Nielsen, 2003; Leung, 2004). For photonic quantum computers, this is vital because photons do not interact with each other and hence the CNOT gate is impossible with usual linear optics (Knill, Laflamme, and Milburn, 2001).

The CNOT gate can be achieved using the quantum teleportation (Gottesman and Chuang, 1999). The quantum circuit in Figure 1 summarizes the procedure. Below, we analyse the quantum circuit in detail.



Figure 1: A quantum circuit to realize the CNOT gate using quantum teleportation. The first and last qubits are the control and target qubits, respectively, and the result is stored in the third and forth qubits, respectively. Here, the two Bell measurements (between the vertical dashed lines) are implemented through the usual gate operations and measurements in the computational basis. In realistic situations, however, they are achieved (non-deterministically in many cases) using special measurement devices (such as parity measurement).

©1988–2022 Wolfram Research, Inc. All rights reserved. http://reference.wolfram.com/language

First, generate the entanglement in the ancillary quantum register.

In[53]:=

```
entangler = QuantumCircuit[
  LogicalForm[Ket[], S@{2, 3, 4, 5}],
  S[{2, 4}, 6], {CNOT[S[2], S[3]], CNOT[S[4], S[5]]},
  "Separator",
  CNOT[S[3], S[4]]
 ]
```

Out[53]=



In[54]:=

```
chi = Elaborate[entangler];
KetFactor[chi, S@{2, 3}] // LogicalForm
```

Out[55]=

$$\left|0_{S_2}0_{S_3}\right\rangle \otimes \left(\frac{1}{2}\left|0_{S_4}0_{S_5}\right\rangle + \frac{1}{2}\left|1_{S_4}1_{S_5}\right\rangle\right) + \left|1_{S_2}1_{S_3}\right\rangle \otimes \left(\frac{1}{2}\left|0_{S_4}1_{S_5}\right\rangle + \frac{1}{2}\left|1_{S_4}0_{S_5}\right\rangle\right)$$

Here is the core part of the quantum circuit to implement the CNOT gate using quantum teleportation.

In[57]:=

```
core = QuantumCircuit[
  {CNOT[S[1], S[2]], CNOT[S[5], S[6]]},
  S[{1, 5}, 6], Measurement[S[{1, 2, 5, 6}, 3]],
  ControlledU[S[2], S[4, 1]], ControlledU[S[2], S[3, 1]],
  ControlledU[S[1], S[3, 3]],
  ControlledU[S[6], S[4, 1]],
  ControlledU[S[5], S[4, 3]], ControlledU[S[5], S[3, 3]]
 ]
```

Out[57]=



Let us construct a small subroutine to facilitate the test of the above quantum circuit.

In[92]:=

```
qtCNOT[{x_, y_}] := Block[
  {in = Ket[S[1] -> x, S[6] -> y]},
  qc = QuantumCircuit[
    LogicalForm[in, S@{1, 6}],
    {entanglement, "Label" -> Ket["χ"]},
    core
   ];
  Echo[qc];
  Table[LogicalForm[KetDrop[Elaborate[qc], S@{1, 2, 5, 6}], S@{3, 4}], 20]
 ]
```

©1988–2022 Wolfram Research, Inc. All rights reserved. http://reference.wolfram.com/language

In[93]:=

```
qtCNOT[{0, 0}]
```

»



Out[93]=

$\{|0_{S_3}0_{S_4}\rangle, |0_{S_3}0_{S_4}\rangle, |0_{S_3}0_{S_4}\rangle, |0_{S_3}0_{S_4}\rangle, |0_{S_3}0_{S_4}\rangle, -|0_{S_3}0_{S_4}\rangle, |0_{S_3}0_{S_4}\rangle, |0_{S_3}0_{S_4}\rangle, |0_{S_3}0_{S_4}\rangle, -|0_{S_3}0_{S_4}\rangle, |0_{S_3}0_{S_4}\rangle, |0_{S_3}0_{S_4}\rangle, |0_{S_3}0_{S_4}\rangle, |0_{S_3}0_{S_4}\rangle, -|0_{S_3}0_{S_4}\rangle, |0_{S_3}0_{S_4}\rangle, |0_{S_3}0_{S_4}\rangle, |0_{S_3}0_{S_4}\rangle, |0_{S_3}0_{S_4}\rangle, |0_{S_3}0_{S_4}\rangle\}$

In[94]:=

```
qtCNOT[{0, 1}]
```

»



Out[94]=

$\{|0_{S_3}1_{S_4}\rangle, |0_{S_3}1_{S_4}\rangle, |0_{S_3}1_{S_4}\rangle, |0_{S_3}1_{S_4}\rangle, -|0_{S_3}1_{S_4}\rangle, |0_{S_3}1_{S_4}\rangle, |0_{S_3}1_{S_4}\rangle, |0_{S_3}1_{S_4}\rangle, |0_{S_3}1_{S_4}\rangle, -|0_{S_3}1_{S_4}\rangle, |0_{S_3}1_{S_4}\rangle, |0_{S_3}1_{S_4}\rangle, |0_{S_3}1_{S_4}\rangle, -|0_{S_3}1_{S_4}\rangle, -|0_{S_3}1_{S_4}\rangle, |0_{S_3}1_{S_4}\rangle, -|0_{S_3}1_{S_4}\rangle, |0_{S_3}1_{S_4}\rangle, |0_{S_3}1_{S_4}\rangle, |0_{S_3}1_{S_4}\rangle\}$

©1988–2022 Wolfram Research, Inc. All rights reserved. http://reference.wolfram.com/language

In[95]:=

```
qtCNOT[{1, 0}]
```

»



Out[95]=

$\{|1_{S_3}1_{S_4}\rangle, |1_{S_3}1_{S_4}\rangle, |1_{S_3}1_{S_4}\rangle, |1_{S_3}1_{S_4}\rangle, |1_{S_3}1_{S_4}\rangle, -|1_{S_3}1_{S_4}\rangle, |1_{S_3}1_{S_4}\rangle, -|1_{S_3}1_{S_4}\rangle, |1_{S_3}1_{S_4}\rangle, |1_{S_3}1_{S_4}\rangle,$
$|1_{S_3}1_{S_4}\rangle, |1_{S_3}1_{S_4}\rangle, |1_{S_3}1_{S_4}\rangle, |1_{S_3}1_{S_4}\rangle, |1_{S_3}1_{S_4}\rangle, -|1_{S_3}1_{S_4}\rangle, |1_{S_3}1_{S_4}\rangle, |1_{S_3}1_{S_4}\rangle, |1_{S_3}1_{S_4}\rangle, |1_{S_3}1_{S_4}\rangle\}$

In[96]:=

```
qtCNOT[{1, 1}]
```

»



Out[96]=

$\{|1_{S_3}0_{S_4}\rangle, -|1_{S_3}0_{S_4}\rangle, |1_{S_3}0_{S_4}\rangle, |1_{S_3}0_{S_4}\rangle, |1_{S_3}0_{S_4}\rangle, |1_{S_3}0_{S_4}\rangle, -|1_{S_3}0_{S_4}\rangle, |1_{S_3}0_{S_4}\rangle, |1_{S_3}0_{S_4}\rangle, |1_{S_3}0_{S_4}\rangle,$
$|1_{S_3}0_{S_4}\rangle, |1_{S_3}0_{S_4}\rangle, |1_{S_3}0_{S_4}\rangle, |1_{S_3}0_{S_4}\rangle, |1_{S_3}0_{S_4}\rangle, |1_{S_3}0_{S_4}\rangle, |1_{S_3}0_{S_4}\rangle, |1_{S_3}0_{S_4}\rangle, -|1_{S_3}0_{S_4}\rangle, -|1_{S_3}0_{S_4}\rangle\}$

Related Guides

- Quantum Computation and Information

Related Tech Notes

- Quantum Algorithms
- QuantumCircuit Usage Examples
- Quantum Computation and Information: Overview

©1988–2022 Wolfram Research, Inc. All rights reserved. http://reference.wolfram.com/language

Related Wolfram Education Group Courses

An Elementary Introduction to the Wolfram Language

The Wolfram Language: Fast Introduction for Programmers

Related Links

- C. H. Bennet, G. Brassard, C. Crepeau, R. Jozsa, A. Peres, and W. K. Wootters, Physical Review Letters 70, 1895 (1993) , “Teleporting an Unknown Quantum State via Dual Classical and Einstein–Podolsky–Rosen Channels.”
- C. H. Bennett and S. J. Wiesner, Physical Review Letters 69 (20), 2881 (1992) , “Communication via one– and two–particle operators on Einstein–Podolsky–Rosen states.”
- D. Gottesman and I. Chuang, Nature 402, 390–393 (1999) , "Quantum teleportation as a universal computational primitive."
- M. A. Nielsen. Physics Letters A, 308, 96–100 (2003) , “Quantum computation by measurement and quantum memory.”
- D. W. Leung, International Journal of Quantum Information 2, 33–43, (2004) , “Quantum computation by measurements.”
- E. Knill, R. Laflamme, and G. J. Bilburn, Nature 409, 46 (2001) , "A scheme for efficient quantum computation with linear optics."
- M. Nielsen and I. L. Chuang (2022) , Quantum Computation and Quantum Information (Cambridge University Press, 2011).
- Mahn–Soo Choi (2022) , A Quantum Computation Workbook (Springer, 2022).

©1988–2022 Wolfram Research, Inc. All rights reserved. http://reference.wolfram.com/language