Q3 TUTORIAL

Related Tech Notes
Related Guides
Functions
URL

Demo: Kitaev Chain

DATED IN 13.1

The Kitaev chain is a prototype example of the topological superconductor. Here, we demonstrate how to examine its properties Kitaev (2001), Kitaev's original, for technical details of the model.

FockNumber or Q	Number operator
FockHopping or Hop	Constructs the hopping Hamiltonian terms
FockPairing or Pair	Constructs the pairing Hamiltonian terms

Some useful functions to study the Kitaev chain.

Load the Fock package.

```
In[1]:= << Q3`
```

Let c denote the (Dirac) Fermion operator.

```
In[2]:= Let[Fermion, c]
```

Let t, μ, and Δ denote the hopping amplitude, chemical potential, and pairing potential, respectively.

```
In[3]:= Let[Real, t, μ, Δ]
```

The number of sites of the one-dimensional lattice.

```
In[4]:= $L = 8;
```

Define some Lists of operators to be used later.

```
In[5]:= cc = c[Range[1, $L]]
        cccc = Join[cc, Dagger@cc]
```

Out[5]= $\{c_1, c_2, c_3, c_4, c_5, c_6, c_7, c_8\}$

Out[6]= $\{c_1, c_2, c_3, c_4, c_5, c_6, c_7, c_8, c_1^\dagger, c_2^\dagger, c_3^\dagger, c_4^\dagger, c_5^\dagger, c_6^\dagger, c_7^\dagger, c_8^\dagger\}$

Now constructs the model Hamiltonian.

In[7]:=
```
H0 = -μ*Q[cc]
Hhop = -t*PlusDagger @ FockHopping[cc]
Hpair = -Δ*PlusDagger @ FockPairing[cc]
Htotal = H0 + Hhop + Hpair
```

Out[7]=
$$-\mu\left(c_1^\dagger c_1 + c_2^\dagger c_2 + c_3^\dagger c_3 + c_4^\dagger c_4 + c_5^\dagger c_5 + c_6^\dagger c_6 + c_7^\dagger c_7 + c_8^\dagger c_8\right)$$

Out[8]=
$$-t\left(c_1^\dagger c_2 + c_2^\dagger c_1 + c_2^\dagger c_3 + c_3^\dagger c_2 + c_3^\dagger c_4 + c_4^\dagger c_3 + c_4^\dagger c_5 + c_5^\dagger c_4 + c_5^\dagger c_6 + c_6^\dagger c_5 + c_6^\dagger c_7 + c_7^\dagger c_6 + c_7^\dagger c_8 + c_8^\dagger c_7\right)$$

Out[9]=
$$-\Delta\left(-c_2 c_1 - c_3 c_2 - c_4 c_3 - c_5 c_4 - c_6 c_5 - c_7 c_6 - c_8 c_7 - c_1^\dagger c_2^\dagger - c_2^\dagger c_3^\dagger - c_3^\dagger c_4^\dagger - c_4^\dagger c_5^\dagger - c_5^\dagger c_6^\dagger - c_6^\dagger c_7^\dagger - c_7^\dagger c_8^\dagger\right)$$

Out[10]=
$$-\Delta\left(-c_2 c_1 - c_3 c_2 - c_4 c_3 - c_5 c_4 - c_6 c_5 - c_7 c_6 - c_8 c_7 - c_1^\dagger c_2^\dagger - c_2^\dagger c_3^\dagger - c_3^\dagger c_4^\dagger - c_4^\dagger c_5^\dagger - c_5^\dagger c_6^\dagger - c_6^\dagger c_7^\dagger - c_7^\dagger c_8^\dagger\right) -$$
$$t\left(c_1^\dagger c_2 + c_2^\dagger c_1 + c_2^\dagger c_3 + c_3^\dagger c_2 + c_3^\dagger c_4 + c_4^\dagger c_3 + c_4^\dagger c_5 + c_5^\dagger c_4 + c_5^\dagger c_6 + c_6^\dagger c_5 + c_6^\dagger c_7 + c_7^\dagger c_6 + c_7^\dagger c_8 + c_8^\dagger c_7\right) -$$
$$\mu\left(c_1^\dagger c_1 + c_2^\dagger c_2 + c_3^\dagger c_3 + c_4^\dagger c_4 + c_5^\dagger c_5 + c_6^\dagger c_6 + c_7^\dagger c_7 + c_8^\dagger c_8\right)$$

Let us take a look at the connectivity of the model. Here, the black thin lines indicate single-particle tunneling and the red thick lines represent the p-wave pairing.

In[11]:=
```
GraphForm[Htotal]
```

Out[11]=



To investigate the quasi-particle spectrum (by means of the Bogoliubov-de Gennes equation), obtain the single-particle BdG Hamiltonian.

In[12]:=
```
matK = CoefficientTensor[Htotal, Dagger@cccc, cccc];
matK // MatrixForm
```

Out[13]//MatrixForm=
$$\begin{pmatrix}
-\frac{\mu}{2} & -\frac{t}{2} & 0 & 0 & 0 & 0 & 0 & 0 & 0 & \frac{\Delta}{2} & 0 & 0 & 0 & 0 & 0 & 0 \\
-\frac{t}{2} & -\frac{\mu}{2} & -\frac{t}{2} & 0 & 0 & 0 & 0 & 0 & -\frac{\Delta}{2} & 0 & \frac{\Delta}{2} & 0 & 0 & 0 & 0 & 0 \\
0 & -\frac{t}{2} & -\frac{\mu}{2} & -\frac{t}{2} & 0 & 0 & 0 & 0 & 0 & -\frac{\Delta}{2} & 0 & \frac{\Delta}{2} & 0 & 0 & 0 & 0 \\
0 & 0 & -\frac{t}{2} & -\frac{\mu}{2} & -\frac{t}{2} & 0 & 0 & 0 & 0 & 0 & -\frac{\Delta}{2} & 0 & \frac{\Delta}{2} & 0 & 0 & 0 \\
0 & 0 & 0 & -\frac{t}{2} & -\frac{\mu}{2} & -\frac{t}{2} & 0 & 0 & 0 & 0 & 0 & -\frac{\Delta}{2} & 0 & \frac{\Delta}{2} & 0 & 0 \\
0 & 0 & 0 & 0 & -\frac{t}{2} & -\frac{\mu}{2} & -\frac{t}{2} & 0 & 0 & 0 & 0 & 0 & -\frac{\Delta}{2} & 0 & \frac{\Delta}{2} & 0 \\
0 & 0 & 0 & 0 & 0 & -\frac{t}{2} & -\frac{\mu}{2} & -\frac{t}{2} & 0 & 0 & 0 & 0 & 0 & -\frac{\Delta}{2} & 0 & \frac{\Delta}{2} \\
0 & 0 & 0 & 0 & 0 & 0 & -\frac{t}{2} & -\frac{\mu}{2} & 0 & 0 & 0 & 0 & 0 & 0 & -\frac{\Delta}{2} & 0 \\
0 & -\frac{\Delta}{2} & 0 & 0 & 0 & 0 & 0 & 0 & \frac{\mu}{2} & \frac{t}{2} & 0 & 0 & 0 & 0 & 0 & 0 \\
\frac{\Delta}{2} & 0 & -\frac{\Delta}{2} & 0 & 0 & 0 & 0 & 0 & \frac{t}{2} & \frac{\mu}{2} & \frac{t}{2} & 0 & 0 & 0 & 0 & 0 \\
0 & \frac{\Delta}{2} & 0 & -\frac{\Delta}{2} & 0 & 0 & 0 & 0 & 0 & \frac{t}{2} & \frac{\mu}{2} & \frac{t}{2} & 0 & 0 & 0 & 0 \\
0 & 0 & \frac{\Delta}{2} & 0 & -\frac{\Delta}{2} & 0 & 0 & 0 & 0 & 0 & \frac{t}{2} & \frac{\mu}{2} & \frac{t}{2} & 0 & 0 & 0 \\
0 & 0 & 0 & \frac{\Delta}{2} & 0 & -\frac{\Delta}{2} & 0 & 0 & 0 & 0 & 0 & \frac{t}{2} & \frac{\mu}{2} & \frac{t}{2} & 0 & 0 \\
0 & 0 & 0 & 0 & \frac{\Delta}{2} & 0 & -\frac{\Delta}{2} & 0 & 0 & 0 & 0 & 0 & \frac{t}{2} & \frac{\mu}{2} & \frac{t}{2} & 0 \\
0 & 0 & 0 & 0 & 0 & \frac{\Delta}{2} & 0 & -\frac{\Delta}{2} & 0 & 0 & 0 & 0 & 0 & \frac{t}{2} & \frac{\mu}{2} & \frac{t}{2} \\
0 & 0 & 0 & 0 & 0 & 0 & \frac{\Delta}{2} & 0 & 0 & 0 & 0 & 0 & 0 & 0 & \frac{t}{2} & \frac{\mu}{2}
\end{pmatrix}$$

Now, study the quasi-particle energy spectrum of the model. Be aware of the finite size effects.

In[14]:=
```
KK[a_, b_, c_] := Block[{μ = a, t = b, Δ = c}, SparseArray@ArrayRules@matK]
```

In[17]:=

```
Manipulate[
 Module[
  {eval = Table[Sort@Eigenvalues[KK[μ, 1., Δ]], {μ, 0, 4, 0.02}]},
  ListLinePlot[Transpose@eval,
   Axes → None,
   Frame → True,
   FrameLabel → {"μ / t", "Eenergy / t"},
   DataRange → {0, 4}]
 ],
 {{Δ, 1}, 0.5, 2, 0.1},
 SaveDefinitions -> True]
```

Out[17]=



Related Guides

- Quantum Many-Body Systems

Related Tech Notes

- Quantum Many-Body Systems with Q3
- Q3: Quick Start

Related Links

- A. Kitaev, Physics-Uspekhi 44, 131 (2001) , "Unpaired Majorana fermions in quantum wires".
- M. Nielsen and I. L. Chuang (2022) , Quantum Computation and Quantum Information (Cambridge University Press, 2011).
- Mahn-Soo Choi (2022) , A Quantum Computation Workbook (Springer, 2022).